

Fun and Games with the Commodore 64

Micheal H. McCabe, August 29, 2015:

Late last week, I discovered that my Educator 64 (a somewhat rare C64 model) had suffered corrosion and rodent damage while in storage. The Educator 64 was an unusual variation of the C64 where a “breadbox” style motherboard was installed in a Commodore PET case with monochrome monitor to create a ruggedized version of the C64 for use in schools. The repair and restoration of this machine might be a good candidate for a future Retrochallenge, so I’ll put that off for now.

Anyway, the discovery of the damaged machine inspired a need to acquire a compatible machine and play a few old-school games while taking a brief survey of my Commodore software and hardware collection.

Checking ebay, I quickly found a candidate system: a classic “breadbox” C64 in its original packaging with power supply. I placed an initial bid of \$25 and waited for the inevitable punks to drive up the price for a few days. As expected, I had to increase my bid to \$50, but there were (surprisingly) no last minute snipers. I won the online auction and paid immediately. The C64 arrived today with no damage.

Here are some pictures of the C64’s first day in its new home with some commentaries on the hardware and software:



Figure 1: Commodore 64 "Breakbox" on the Workbench



Figure 2: Hooked up to the Model 1702 Color Monitor and 1630 Datasette



Figure 3: A Close-up of Vintage Commodore Tape Storage



Figure 4: The 1541 Disk Drive



Figure 5: Splash Screen for a C64 'Star Trek' Game

Playing games was all well and good, but I've never been much of a gamer. The next logical step was to hook up a printer and "dump" several unlabeled tapes to see what programs I might have lurking about. As it turns out, most of the cassette tape programs were actually for the VIC-20, so that machine will be the next one to emerge from storage. In the meantime, the C64 reads the VIC-20 tapes quite well and is able to print the contents, so I'll take this opportunity to make some program listings.

The VIC-1525 Graphic Printer is a nice Seikosha dot-matrix printer that, unfortunately, has become obsolete due to the lack of printer ribbons. The font is quite legible, despite the 7-pin matrix, and the printer makes all the nostalgic noise you would associate with a dot-matrix printer. I'll have to post a short video on Youtube to preserve those sounds for posterity. The thirty-year-old ribbon worked remarkably well, and I obtained several BASIC listings for programs I haven't used since the middle 1980's when I was seduced by the Apple Macintosh.



Figure 6: The VIC 1525 Graphic Printer balanced on the Monitor

Some of the programs were the "sound effect" demonstrations copied from the back of the VIC-20 User's Manual. I'll include some scans of the actual printouts below.

Other programs I located today were simple terminal-emulator programs that I used back then to connect to a time-sharing system at Edinboro University. There were also a few programming assignments from a class I took in BASIC and some simple utility programs to solve various problems in Physics, Chemistry, and Analytic Geometry. If I can overcome the embarrassment of having other people look at my code, I might post them someday.

READY.

```
100 OPEN 5,2,3,CHR$(6)
110 DIM FX(255),TX(255)
200 FOR J=32 TO 64:TX(J)=J:NEXT
210 TX(13)=13:TX(20)=8:RV=18:CT=0
220 FOR J=65 TO 90:K=J+32:TX(J)=K:NEXT
230 FOR J=91 TO 95:TX(J)=J:NEXT
240 FOR J=193 TO 218:K=J-128:TX(J)=K:NEXT
250 TX(146)=16:TX(133)=16
260 FOR J=0 TO 255
270 K=TX(J)
280 IF K<>0 THEN FX(K)=J:FX(K+128)=J
290 NEXT
300 PRINT " ";CHR$(147)
310 GET#5,A$
320 IF A$="" OR ST<>0 THEN 360
330 PRINT " ";CHR$(157);CHR$(FX(ASC(A$)));
340 IF FX(ASC(A$))=34 THEN POKE 212,0
350 GOTO 310
360 PRINT CHR$(RV);" ";CHR$(157);CHR$(146);:GET A$
370 IF A$<>" " THEN PRINT#5,CHR$(TX(ASC(A$)));
380 CT=CT+1
390 IF CT=8 THEN CT=0:RV=164-RV
400 IF (PEEK(37151)AND64)=1 THEN 400
410 GOTO 310
```

READY.

Figure 7: Serial Communications in 1982

READY.

```
1000 REM SET UP AN RS-232 CHANNEL
1010 REM COMM PARAMETERS
1020 REM 300 BAUD
1030 REM 7 BIT ASCII
1040 REM MARK PARITY
1050 REM FULL DUPLEX
1060 OPEN 2,2,3,CHR$(6+32)+CHR$(32+128):REM OPEN THE CHANNEL
1070 GET#2,A$
1080 REM MAIN LOOP
1090 GET B$
1100 IF B$<>" " THEN PRINT#2,B$;
1110 GET#2,C$
1120 PRINT B$;C$;
1130 SR=ST:IF SR=0 THEN 1080
1140 REM ERROR
1150 IF SR AND 1 THEN PRINT "PARITY ERROR."
1160 IF SR AND 2 THEN PRINT "FRAME ERROR."
1170 IF SR AND 4 THEN PRINT "BUFFER OVERFLOW"
1180 IF SR AND 128 THEN PRINT "BREAK ERROR."
1190 IF (PEEK(37151)AND64)=1 THEN 1190
1200 CLOSE 2
1210 END
```

READY.

Figure 8: Documented Code to Open a Serial Channel on the VIC-20

READY.

```
90 PRINT CHR$(147);
100 PRINT "DATA LINK SPEED"
110 PRINT "COMPUTATION:"
120 PRINT
130 INPUT "MBIT/SEC: ";M
135 LET M=M*1016
140 PRINT
150 LET W=M/16
160 LET B=M/8
170 PRINT "XFER MBYTES:";B/1016
180 PRINT "XFER MWORDS:";W/1016
190 PRINT
200 PRINT "1 MINUTE XFER CAPS:"
210 PRINT
220 PRINT "XFER MBYTES:";(B/1016)*60
230 PRINT "XFER MWORDS:";(W/1016)*60
240 PRINT
250 PRINT "1 HOUR XFER CAPS:"
260 PRINT
270 PRINT "XFER MBYTES:";(B/1016)*3600
280 PRINT "XFER MWORDS:";(W/1016)*3600
290 END
```

READY.

Figure 9: Data Link Speed and Capacity for the VIC-20

READY.

```
100 REM MICHEAL H. MCCABE
110 REM RED ALERT
120 POKE 36878,15
130 FOR L=1 TO 10
140 FOR M=180 TO 235 STEP 2
150 POKE 36876,M
160 FOR N=1 TO 10
170 NEXT N
180 NEXT M
190 POKE 36876,0
200 FOR M=1 TO 10
210 NEXT M
220 NEXT L
230 POKE 36878,0
240 END
```

READY.

Figure 10: Red Alert Noises on the VIC-20